Curvature and the Einstein Equation

This is the *Mathematica* notebook *Curvature and the Einstein Equation* available from the book website. From a given metric $g_{\alpha\beta}$, it computes the components of the following: the inverse metric, $g^{\lambda\sigma}$, the Christoffel symbols or affine connection,

$\Gamma^{\lambda}{}_{\mu\nu} = \frac{1}{2} g^{\lambda\sigma}(\partial_{\mu} g_{\sigma\nu} + \partial_{\nu} g_{\sigma\mu} - \partial_{\sigma} g_{\mu\nu})$,

(∂_{α} stands for the partial derivative $\partial / \partial x^{\alpha}$), the Riemann tensor,

$R^{\lambda}{}_{\mu\nu\sigma} = \partial_{\nu} \Gamma^{\lambda}{}_{\mu\sigma} - \partial_{\sigma} \Gamma^{\lambda}{}_{\mu\nu} + \Gamma^{\eta}{}_{\mu\sigma} \Gamma^{\lambda}{}_{\eta\nu} - \Gamma^{\eta}{}_{\mu\nu} \Gamma^{\lambda}{}_{\eta\sigma}$,

the Ricci tensor

$R_{\mu\nu} = R^{\lambda}{}_{\mu\lambda\nu}$,

the scalar curvature,

$R = g^{\mu\nu} R_{\mu\nu}$,

and the Einstein tensor,

$G_{\mu\nu} = R_{\mu\nu} - \frac{1}{2} g_{\mu\nu} R$.

You must input the covariant components of the metric tensor $g_{\mu\nu}$ by editing the relevant input line in this *Mathematica* notebook. You may also wish to change the names of the coordinates. Only the nonzero components of the above quantities are displayed as the output. All the components computed are in the *coordinate basis* in which the metric was specified.

Clearing the values of symbols:

First clear any values that may already have been assigned to the names of the various objects to be calculated. The names of the coordinates that you will use are also cleared.

In[21]:=
```
Clear[coord, metric , inversemetric ,
  affine, riemann , ricci, scalar, einstein, r, θ, ϕ, t]
```

Setting the dimension:

The dimension **n** of the spacetime (or space) must be set:

In[22]:=
```
n = 4
```

Out[22]=
```
4
```

Defining a list of coordinates:

The example given here is the Schwarzschild metric. The coordinate choice of Schwarzschild is appropriate for this spherically symmetric spacetime.

In[23]:=
```
coord = {t, r, θ, ϕ}
```

Out[23]=
```
{t, r, θ, ϕ}
```

You can change the names of the coordinates by simply editing the definition of **coord**, for example, to **coord = {x, y, z, t}**, when another set of coordinate names is more appropriate. In this program indices range over **1** to **n.** Thus for spacetime they range from 1 to 4 and x^4 is the same as x^0 used in the text.

Defining the metric:

Input the metric as a list of lists, i.e., as a matrix. You can input the components of any metric here, but you must specify them as explicit functions of the coordinates.

In[24]:=
```
metric =
  {{-Exp[ν[r]], 0, 0, 0}, {0, Exp[λ[r]], 0, 0}, {0, 0, r^2, 0}, {0, 0, 0, r^2 Sin[θ]^2}}
```

Out[24]= $\{\{-e^{\nu[r]}, 0, 0, 0\}, \{0, e^{\lambda[r]}, 0, 0\}, \{0, 0, r^2, 0\}, \{0, 0, 0, r^2 \mathrm{Sin}[\theta]^2\}\}$

You can also display this in matrix form.

In[25]:=
```
metric // MatrixForm
```

Out[25]//MatrixForm=

$$\begin{pmatrix} -e^{\nu[r]} & 0 & 0 & 0 \\ 0 & e^{\lambda[r]} & 0 & 0 \\ 0 & 0 & r^2 & 0 \\ 0 & 0 & 0 & r^2 \mathrm{Sin}[\theta]^2 \end{pmatrix}$$

Note:

It is important not to use the symbols, **i**, **j**, **k**, **l**, **s**, or **n** as constants or coordinates in the metric that you specify above. The reason is that the first five of those symbols are used as summation or table indices in the calculations done below, and **n** is the dimension of the space. For example, if **m** were used as a summation or table index below, then you would get the wrong answer for the present metric because the **m** in the metric would be treated as an index, rather than as the mass.

Calculating the inverse metric:

The inverse metric is obtained through matrix inversion.

In[26]:=
```
inversemetric = Simplify[Inverse[metric ]]
```

Out[26]= $\left\{\{-e^{-\nu[r]}, 0, 0, 0\}, \{0, e^{-\lambda[r]}, 0, 0\}, \left\{0, 0, \frac{1}{r^2}, 0\right\}, \left\{0, 0, 0, \frac{\mathrm{Csc}[\theta]^2}{r^2}\right\}\right\}$

This can also be displayed in matrix form:

In[27]:=
```
inversemetric // MatrixForm
```

Out[27]//MatrixForm=

$$\begin{pmatrix} -e^{-\nu[r]} & 0 & 0 & 0 \\ 0 & e^{-\lambda[r]} & 0 & 0 \\ 0 & 0 & \frac{1}{r^2} & 0 \\ 0 & 0 & 0 & \frac{\mathrm{Csc}[\theta]^2}{r^2} \end{pmatrix}$$

Calculating the Christoffel symbols:

The calculation of the components of the Christoffel symbols is done by transcribing the definition given earlier into the notation of *Mathematica* and using the *Mathematica* functions **D** for taking partial derivatives, **Sum** for summing over repeated indices, **Table** for forming a list of components, and **Simplify** for simplifying the result.

In[28]:=
```
affine:=affine=Simplify[Table[(1/2)*Sum[(inversemetric[[i, s]])*
            (D[metric[[s, j]], coord[[k]]]+
               D[metric[[s, k]], coord[[j]]]-
               D[metric[[j, k]], coord[[s]]]), {s, 1, n}],
        {i, 1, n}, {j, 1, n}, {k, 1, n}]]
```

Displaying the Christoffel symbols:

The nonzero Christoffel symbols are displayed below. You need not follow the details of constructing the functions that we use for that purpose. In the output the symbol Γ[1,2,3] stands for $\Gamma^1{}_{23}$. Because the Christoffel symbols are symmetric under interchange of the last two indices, only the independent components are displayed.

In[29]:=
```
listaffine:=
  Table[If[UnsameQ[affine[[i, j, k]], 0], {ToString[Γ[i, j, k]], affine[[i, j, k]]}],
    {i, 1, n}, {j, 1, n}, {k, 1, j}]
```

In[30]:=
```
TableForm[Partition[DeleteCases[Flatten[listaffine], Null], 2],
  TableSpacing→{2, 2}]
```

Out[30]//TableForm=

Γ[1, 2, 1]	$\frac{\nu'[r]}{2}$
Γ[2, 1, 1]	$\frac{1}{2} e^{-\lambda[r]+\nu[r]}\, \nu'[r]$
Γ[2, 2, 2]	$\frac{\lambda'[r]}{2}$
Γ[2, 3, 3]	$-e^{-\lambda[r]}\, r$
Γ[2, 4, 4]	$-e^{-\lambda[r]}\, r\, \mathrm{Sin}[\theta]^2$
Γ[3, 3, 2]	$\frac{1}{r}$
Γ[3, 4, 4]	$-\mathrm{Cos}[\theta]\, \mathrm{Sin}[\theta]$
Γ[4, 4, 2]	$\frac{1}{r}$
Γ[4, 4, 3]	$\mathrm{Cot}[\theta]$

Calculating and displaying the Riemann tensor:

The components of the Riemann tensor, $R^\lambda{}_{\mu\nu\sigma}$, are calculated using the definition given above.

In[31]:=
```
riemann:=riemann=Simplify[Table[
        D[affine[[i, j, l]], coord[[k]]]-D[affine[[i, j, k]], coord[[l]]]+
          Sum[
            affine[[s, j, l]] affine[[i, k, s]]-affine[[s, j, k]] affine[[i, l, s]],
            {s, 1, n}],
        {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}]]
```

The nonzero components are displayed by the following functions. In the output, the symbol R[1, 2, 1, 3] stands for $R^1{}_{213}$, and similarly for the other components. You can obtain R[1, 2, 3, 1] from R[1, 2, 1, 3] using the antisymmetry of the Riemann tensor under exchange of the last two indices. The antisymmetry under exchange of the first two indices of $R_{\lambda\mu\nu\sigma}$ is not evident in the output because the components of $R^\lambda{}_{\mu\nu\sigma}$ are displayed.

In[32]:=
```
listriemann:=Table[If[UnsameQ[riemann[[i, j, k, l]], 0],
        {ToString[R[i, j, k, l]], riemann[[i, j, k, l]]}],
      {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, k-1}]
```

In[33]:=
```
TableForm [Partition[DeleteCases[Flatten[listriemann ], Null], 2],
  TableSpacing→ {2, 2}]
```

Out[33]//TableForm=

R[1, 2, 2, 1]	$\frac{1}{4}(-\lambda'[r]\,\nu'[r]+\nu'[r]^2+2\,\nu''[r])$
R[1, 3, 3, 1]	$\frac{1}{2}e^{-\lambda[r]}\,r\,\nu'[r]$
R[1, 4, 4, 1]	$\frac{1}{2}e^{-\lambda[r]}\,r\,\mathrm{Sin}[\theta]^2\,\nu'[r]$
R[2, 1, 2, 1]	$\frac{1}{4}e^{-\lambda[r]+\nu[r]}\,(-\lambda'[r]\,\nu'[r]+\nu'[r]^2+2\,\nu''[r])$
R[2, 3, 3, 2]	$-\frac{1}{2}e^{-\lambda[r]}\,r\,\lambda'[r]$
R[2, 4, 4, 2]	$-\frac{1}{2}e^{-\lambda[r]}\,r\,\mathrm{Sin}[\theta]^2\,\lambda'[r]$
R[3, 1, 3, 1]	$\frac{e^{-\lambda[r]+\nu[r]}\,\nu'[r]}{2\,r}$
R[3, 2, 3, 2]	$\frac{\lambda'[r]}{2\,r}$
R[3, 4, 4, 3]	$(-1+e^{-\lambda[r]})\,\mathrm{Sin}[\theta]^2$
R[4, 1, 4, 1]	$\frac{e^{-\lambda[r]+\nu[r]}\,\nu'[r]}{2\,r}$
R[4, 2, 4, 2]	$\frac{\lambda'[r]}{2\,r}$
R[4, 3, 4, 3]	$1-e^{-\lambda[r]}$

Calculating and displaying the Ricci tensor:

The Ricci tensor $R_{\mu\nu}$ was defined by summing the first and third indices of the Riemann tensor (which has the first index already raised).

In[34]:=
```
ricci:=
  ricci= Simplify [Table[Sum [riemann [[i, j, i, l]], {i, 1, n}], {j, 1, n}, {l, 1, n}]]
```

Next we display the nonzero components. In the output, R[1, 2] denotes R_{12}, and similarly for the other components.

In[35]:=
```
listricci:= Table[If[UnsameQ [ricci[[j, l]], 0],
      {ToString[R[j, l]], ricci[[j, l]]}], {j, 1, n}, {l, 1, j}]
```

In[36]:=
```
TableForm [Partition[DeleteCases[Flatten[listricci], Null], 2],
  TableSpacing→ {2, 2}]
```

Out[36]//TableForm=

R[1, 1]	$\frac{e^{-\lambda[r]+\nu[r]}\,((4-r\,\lambda'[r])\,\nu'[r]+r\,\nu'[r]^2+2\,r\,\nu''[r])}{4\,r}$
R[2, 2]	$\frac{\lambda'[r]\,(4+r\,\nu'[r])-r\,(\nu'[r]^2+2\,\nu''[r])}{4\,r}$
R[3, 3]	$\frac{1}{2}e^{-\lambda[r]}\,(-2+2\,e^{\lambda[r]}+r\,\lambda'[r]-r\,\nu'[r])$
R[4, 4]	$\frac{1}{2}e^{-\lambda[r]}\,\mathrm{Sin}[\theta]^2\,(-2+2\,e^{\lambda[r]}+r\,\lambda'[r]-r\,\nu'[r])$

A vanishing table (as with the Schwarzschild metric example) means that the vacuum Einstein equation is satisfied.

Calculating the scalar curvature:

The scalar curvature R is calculated using the inverse metric and the Ricci tensor. The result is displayed in the output line.

In[37]:=
```
scalar = Simplify [Sum [inversemetric [[i, j]] ricci[[i, j]], {i, 1, n}, {j, 1, n}]]
```

Out[37]=
$$\frac{e^{-\lambda[r]}\left(-4+4 e^{\lambda[r]}-4 r \nu'[r]-r^2 \nu'[r]^2+r \lambda'[r](4+r \nu'[r])-2 r^2 \nu''[r]\right)}{2 r^2}$$

Calculating the Einstein tensor:

The Einstein tensor, $G_{\mu\nu} = R_{\mu\nu} - \frac{1}{2} g_{\mu\nu} R$, is found from the tensors already calculated.

In[38]:=
```
einstein := einstein = Simplify [ricci - (1 / 2) scalar * metric ]
```

The results are displayed in the same way as for the Ricci tensor earlier.

In[39]:=
```
listeinstein := Table[If[UnsameQ [einstein[[j, l]], 0],
      {ToString[G[j, l]], einstein[[j, l]]}] , {j, 1, n}, {l, 1, j}]
```

In[40]:=
```
TableForm [Partition[DeleteCases[Flatten[listeinstein], Null], 2],
  TableSpacing → {2, 2}]
```

Out[40]//TableForm=

G[1, 1]	$\frac{e^{-\lambda[r]+\nu[r]}\left(-1+e^{\lambda[r]}+r \lambda'[r]\right)}{r^2}$
G[2, 2]	$\frac{1-e^{\lambda[r]}+r \nu'[r]}{r^2}$
G[3, 3]	$\frac{1}{4} e^{-\lambda[r]} r\left(2 \nu'[r]+r \nu'[r]^2-\lambda'[r](2+r \nu'[r])+2 r \nu''[r]\right)$
G[4, 4]	$\frac{1}{4} e^{-\lambda[r]} r \operatorname{Sin}[\theta]^2\left(2 \nu'[r]+r \nu'[r]^2-\lambda'[r](2+r \nu'[r])+2 r \nu''[r]\right)$

A vanishing table means that the vacuum Einstein equation is satisfied!

Acknowledgment

This program was kindly written by *Leonard Parker, University of Wisconsin, Milwaukee* especially for this text.